Use of Maple in Determining Radial Eigenfunctions for the Bound-State Aharonov-Bohm Effect

W. P. Healy
Department of Mathematics
RMIT University
liam.healy@rmit.edu.au

H. Hung
Department of Mathematics
RMIT University

Abstract

Numerical and graphical representations of radial eigenfunctions for a charged particle contained in a cylindrical shell are obtained by using Maple. The general solution of the radial Schrödinger equation can be expressed analytically in terms of Bessel functions and confluent hypergeometric functions when there is no magnetic-induction field and a constant longitudinal magnetic-induction field, respectively, in the shell. In both cases, the transcendental equation arising from imposition of the boundary conditions is solved numerically to obtain approximations to the eigenvalues. The bound-state Aharonov-Bohm effect is exhibited through the change in the eigenfunctions due to a longitudinal magnetic-induction field confined within the inner cylinder, which the charged particle cannot penetrate. This paper is partly based on a fourth-year (Honours) student project and illustrates some of the numerical and graphical capabilities that modern computer algebra systems such as Maple make available to students.

1 Introduction

The Aharonov-Bohm effect [1] is a fundamental but surprising feature of quantum mechanics in which a charged particle is influenced by a static magnetic-induction field that is present in a region of space from which the particle is rigorously excluded. The configuration space available to the particle must be multiply connected and must encircle the region of confined magnetic flux. The effect has been verified experimentally by Tonomura *et al.* (see [2]) by diffracting electrons around a toroidal ferromagnet coated with superconducting material. In these scattering experiments, the effect is manifest as a shift in the observed electron interference pattern caused by the inaccessible magnetic-induction field. The Aharonov-Bohm effect is also predicted [3, 4] to occur for a charged particle bound in a multiply connected region, such as a cylindrical shell, that is threaded by a magnetic-induction field. The bound-state effect consists of a shift in the discrete energy levels and a change in the energy eigenfunctions and associated position probability densities. Both the scattering and bound-state effects are periodic in the confined magnetic flux with period equal to London's natural unit of flux [4]. (For an electron, the magnitude of the unit of flux is approximately 4.1×10^{-7} Mx in Gaussian units and 4.1×10^{-15} Wb in SI units.)

Exact general solutions of the radial part of the time-independent Schrödinger equation (see Section 2) for a charged particle in a cylindrical shell can be found when (a) there is a cylindrically symmetric static longitudinal magnetic-induction field inside the inner cylinder and no field in the shell or (b) there is a constant non-zero longitudinal field in the shell as well as an inner field as in (a). Even in these cases, however, determination of the radial eigenvalues and eigenfunctions by imposition of the boundary conditions at the inner and outer walls of the cylindrical shell must rely on numerical methods, although an asymptotic formula that gives approximations to the energy eigenvalues in the domain of large radial quantum numbers has been derived recently [5]. In this paper, we describe the use of Maple to obtain numerical and graphical representations of radial eigenfunctions in both cases (a) and (b) above. The bound-state Aharonov-Bohm effect is exhibited through the change in the eigenfunctions caused by the confined field. This change is small (due to the periodicity of the effect and the smallness of London's unit) but nevertheless visible. The eigenfunctions for case (a) are expressible in terms of Bessel functions and are considered in Section 3 while those for case (b) are expressible in terms of confluent hypergeometric functions and are considered in Section 4.

2 Radial Schrödinger Equation

The bound-state system consists of a particle of mass m and charge e confined by perfectly reflecting walls to the cylindrical shell S defined in terms of cylindrical polar coordinates (ρ, θ, z) by $a \leq \rho \leq b$ and $0 \leq z \leq d$, where a, b and d are positive constants. The flux f of the cylindrically symmetric longitudinal magnetic-induction field *inside* the inner cylinder $\rho = a$ will be measured in London's unit hc/e, in which h is Planck's constant, c is the speed of light *in vacuo* and Gaussian units are used. The θ- and z-dependent parts of the energy eigenfunction for a given stationary state are independent of f but the radial part $R_{n\lambda}(\rho)$ is flux dependent. It satisfies the radial Schrödinger equation

$$\rho^2 R''_{n\lambda} + \rho R'_{n\lambda} + \left[\alpha^2_{n\lambda}\rho^2 - \left(\lambda - \frac{2\pi e}{hc}\rho A_0\right)^2\right] R_{n\lambda} = 0. \tag{1}$$

Here the positive integer n is the radial quantum number and $\lambda = l - f$, where the integer l is the canonical angular-momentum quantum number. Also $A_0(\rho)\hat{\boldsymbol{\theta}}$ is the vector potential in the Coulomb gauge due to any cylindrically symmetric longitudinal magnetic-induction field *outside* the inner cylinder $\rho = a$. If $\rho \geq a$,

$$A_0(\rho) = \frac{1}{\rho}\int_a^\rho B(\rho')\rho'\, d\rho' \tag{2}$$

where $B(\rho)\hat{\boldsymbol{z}}$ is the magnetic-induction field. The allowed values of the positive constants $\alpha_{n\lambda}$ are determined by the boundary conditions

$$R_{n\lambda}(a) = 0 = R_{n\lambda}(b) \tag{3}$$

and the eigenfunctions $R_{n\lambda}$ will be normalized on $[a, b]$ with respect to the weight function ρ. Only values of f in the interval $[0, 1)$ need be considered, as the eigenvalue spectrum and the set of eigenstates are invariant under the addition of an arbitrary integer to f.

3 Zero Field in Shell

If there is no magnetic-induction field in S, the function A_0 is identically zero and the radial Schrödinger equation (1) reduces to Bessel's equation of order λ with parameter $\alpha_{n\lambda}$. An unnormalized eigenfunction is given by

$$J_\lambda(\alpha_{n\lambda}a)Y_\lambda(\alpha_{n\lambda}\rho) - Y_\lambda(\alpha_{n\lambda}a)J_\lambda(\alpha_{n\lambda}\rho). \tag{4}$$

This obviously satisfies the boundary condition at $\rho = a$ and will satisfy the boundary condition at $\rho = b$ also if $\alpha_{n\lambda}$ $(n = 1, 2, \ldots)$ is a positive root of the equation

$$\Delta(\alpha) = J_\lambda(\alpha a)Y_\lambda(\alpha b) - Y_\lambda(\alpha a)J_\lambda(\alpha b) = 0 \tag{5}$$

in which a, b and λ are fixed. The following Maple code shows how to determine the allowed value of α numerically for the case in which $b = 11a$,

$n = 4$, $l = 5$ and $f = 0$ (and hence $\lambda = l - f = 5$). The radial distance ρ will be expressed in units of a. $\Delta(\alpha)$ is first plotted (see Figure 1) in order to determine intervals in which its zeros occur.

```
> alias(J=BesselJ,Y=BesselY):
> Delta:=J(5,alpha)*Y(5,11*alpha)-Y(5,alpha)*J(5,11*alpha );
```

$$\Delta := \mathrm{J}(5,\,\alpha)\,\mathrm{Y}(5,\,11\,\alpha) - \mathrm{Y}(5,\,\alpha)\,\mathrm{J}(5,\,11\,\alpha)$$

```
> plot(Delta,alpha=0.77..2,color=black,numpoints=200);
```



Figure 1: Graph of expression $\Delta(\alpha)$

Approximations to the allowed value α_{45} and the normalized eigenfunction R_{45} are now obtained. These are assigned to the Maple variables `A` and `RA`, respectively. The graph of R_{45} is then plotted.

```
> A:=fsolve(Delta,alpha,1.7..1.8);
```

$$A := 1.725483795$$

```
> R:=J(5,A)*Y(5,A*rho)-Y(5,A)*J(5,A*rho):
> N:=evalf(Int(rho*R^2,rho=1..11));
```

$$N := 735.6919769$$

```
> RA:=R/sqrt(N):
> plot(RA,rho=1..11,color=black,numpoints=200);
```

The approximate expression for the normalized eigenfunction R_{45} obtained from the Maple code above is

$$0.7145697131 J_5(1.725483795\rho) + 0.0001295721657 Y_5(1.725483795\rho) \quad (6)$$



Figure 2: Radial eigenfunction for $n = 4$ and $f = 0$

and the graph of this expression is shown in Figure 2. Since $f = 0$, R_{45} is an eigenfunction for a state in which the confined magnetic-induction field is zero.

Approximations to $\alpha_{4,4.6}$ and the eigenfunction $R_{4,4.6}$ corresponding to the same values of b, n and l as above but with $f = 0.4$ (and hence with $\lambda = 4.6$) are obtained next. These are assigned to the variables `B` and `RB`, respectively.

```
> Delta:=J(4.6,alpha)*Y(4.6,11*alpha)-
> Y(4.6,alpha)*J(4.6,11*alpha);
```

$$\Delta := \mathrm{J}(4.6,\, \alpha)\, \mathrm{Y}(4.6,\, 11\,\alpha) - \mathrm{Y}(4.6,\, \alpha)\, \mathrm{J}(4.6,\, 11\,\alpha)$$

```
> B:=fsolve(Delta,alpha=1.6..1.7);
```

$$B := 1.676189326$$

```
> R:=J(4.6,B)*Y(4.6,B*rho)-Y(4.6,B)*J(4.6,B*rho ):
> N:=evalf(Int(rho*R^2,rho=1..11));
```

$$N := 280.3419663$$

```
> RB:=R/sqrt(N):
> plot({RA,RB},rho=1..11,color=black,numpoints=200);
> plot(RB-RA,rho=1..11,color=black,numpoints=200);
> with(plots,cylinderplot):
> cylinderplot([rho,theta,RB-RA],rho=1..11,
> theta=Pi/2..2*Pi,style=patch,axes=frame);
```



Figure 3: Radial eigenfunctions for $f = 0$ and $f = 0.4$



Figure 4: Difference of radial eigenfunctions for $f = 0.4$ and $f = 0$



Figure 5: Cylinderplot exhibiting the Aharonov-Bohm effect

The approximate expression for the normalized eigenfunction $R_{4,4.6}$ is

$$0.7029970589J_{4.6}(1.676189326\rho) + 0.0003793551008Y_{4.6}(1.676189326\rho). \quad (7)$$

R_{45} and $R_{4,4.6}$ are plotted together in Figure 3. The difference $R_{4,4.6} - R_{45}$ is plotted in Figure 4 and is also shown in Figure 5 as a 'cylinderplot' with one quadrant excised. Figures 3 to 5 exhibit the bound-state Aharonov-Bohm effect by showing the difference between the eigenfunctions $R_{4,4.6}$ and R_{45}. This difference is due solely to the change from 0 to 0.4 in the value of the flux f confined within the inner cylinder.

4 Constant Non-Zero Field in Shell

If there is a constant non-zero longitudinal magnetic-induction field in S, then Equation (2) implies that

$$\frac{2\pi e}{hc}\rho A_0 = f'\left[\left(\frac{\rho}{a}\right)^2 - 1\right] \quad (8)$$

where f' is the scaled flux (in units of hc/e) that the field in the shell would have if this field were inside the inner cylinder. It will be assumed here that the difference between the fluxes f and f' is not an integer and that the sense of the z axis is chosen so that $f' > 0$. The general solution of Equation (1) can then be expressed in terms of the confluent hypergeometric function F, which has two parameters and one argument [6]. The eigenfunction $R_{n\lambda}$ is proportional to

$$\exp\left(-\frac{f'}{2a^2}\rho^2\right)\left[\left(\frac{\rho}{a}\right)^\mu F_2(\beta_{n\lambda},a)F_1(\beta_{n\lambda},\rho)-\left(\frac{a}{\rho}\right)^\mu F_1(\beta_{n\lambda},a)F_2(\beta_{n\lambda},\rho)\right] \tag{9}$$

where the functions F_1 and F_2 are defined by

$$F_1(\beta,\rho)=F\left(-\beta+\frac{1}{2}\mu+\frac{1}{2},1+\mu,\frac{f'}{a^2}\rho^2\right) \tag{10}$$

and

$$F_2(\beta,\rho)=F\left(-\beta-\frac{1}{2}\mu+\frac{1}{2},1-\mu,\frac{f'}{a^2}\rho^2\right). \tag{11}$$

Also $\beta_{n\lambda}$ is given in terms of $\alpha_{n\lambda}$ by

$$\beta_{n\lambda}=\frac{a^2}{4f'}\alpha_{n\lambda}^2+\frac{1}{2}\mu \tag{12}$$

and $\mu=\lambda+f'$. Expression (9) obviously satisfies the boundary condition at $\rho=a$ and will satisfy the boundary condition at $\rho=b$ also if $\beta_{n\lambda}$ is a root of the equation

$$\Delta(\beta)=\left(\frac{b}{a}\right)^\mu F_2(\beta,a)F_1(\beta,b)-\left(\frac{a}{b}\right)^\mu F_1(\beta,a)F_2(\beta,b)=0. \tag{13}$$

It should be noted from Equation (12) that $\beta_{n\lambda}$ cannot be less than $\mu/2$.

The eigenfunctions R_{25} and $R_{2,4.6}$ will now be plotted together with a as the radial unit of length. For both eigenfunctions, $b=11a$, $n=2$, $l=5$ and $f'=0.3$. For R_{25}, $f=0$ and hence $\lambda=5$ and $\mu=5.3$.

```
> readlib(hypergeom):  alias(F=hypergeom):
> F1:=(beta,rho)->F([-beta+3.15],[6.3],0.3*rho^ 2):
> F2:=(beta,rho)->F([-beta-2.15],[-4.3],0.3*rho ^2):
> Delta:=11^5.3*F2(beta,1)*F1(beta,11)-
> (1/11)^5.3*F1(beta,1)*F2(beta,11):
```

It was determined graphically that the second allowed value of β for this case lies in the interval $(4.1, 4.2)$.

```
> A:=fsolve(Delta,beta,4.1..4.2);
```

$$A := 4.150170607$$

```
> R:=exp(-0.15*rho^2)*(rho^5.3*F2(A,1)*F1(A,rho)-
> rho^(-5.3)*F1(A,1)*F2(A,rho)):
> N:=evalf(Int(rho*R^2,rho=1..11));
```

$$N := 77676.10260$$

```
> RA:=R/sqrt(N):
```

The Maple code to determine $R_{2,4.6}$ is the same as that used to obtain R_{25} except that now $f = 0.4$ and hence $\lambda = 4.6$ and $\mu = 4.9$. The expression for $R_{2,4.6}$ was assigned to the Maple variable `RB` and both `RA` and `RB` were plotted. The graph is shown in Figure 6. This illustrates the Aharonov-Bohm effect



Figure 6: Radial eigenfunctions for $n = 2$ and $f' = 0.3$

when there is a constant field (with $f' = 0.3$) in the shell and the scaled flux f of the field inside the inner cylinder changes from 0 to 0.4.

5 Discussion

This work is partly based on a fourth-year (Honours) student project completed in 1997. Numerical and graphical representations of radial eigenfunctions for a charged particle contained in a cylindrical shell have been obtained by using Maple. The work has included the numerical solution of transcendental equations involving special mathematical functions (Bessel functions and confluent hypergeometric functions) as well as the plotting of these functions. Maple's graphical capabilities have enabled easy visualization of a subtle feature of quantum mechanics — the bound-state Aharonov-Bohm effect. Such visualization of solutions of differential equations is a major benefit of the use of Maple in the mathematics curriculum.

References

[1] Y. Aharonov and D. Bohm, Phys. Rev. **115**, 485 (1959).

[2] M. Peshkin and A. Tonomura, *The Aharonov-Bohm Effect* (Springer-Verlag, Berlin, 1989).

[3] M. Peshkin, Phys. Rev. A **23**, 360 (1981).

[4] M. Peshkin, Phys. Rep. **80**, 375 (1981).

[5] R. A. Samandra and W. P. Healy (submitted for publication)

[6] L. D. Landau and E. M. Lifshitz, *Quantum Mechanics: Non-Relativistic Theory* (Pergamon, Oxford, 1977).